��Licence Economie et Gestion

Outils Informatiques 2

Responsables : C. Hanen – R. Chevallier ��

Corrigés de quelques exercices



Exercice 3 (TD 3)



Soit m la moyenne des notes d’un étudiant donnée par l'utilisateur. Déterminer la décision du jury (ajourné, admis passable, admis AB, admis B, admis TB).



Corrigé

Sub Resultat()

	Dim note As Double

	note=InputBox("Entrez votre note")

	If note < 10 Then MsgBox("Ajourne(e)")

	If note>=10 And note<12 Then MsgBox("Admis Passable")

	If note>=12 And note<14 Then MsgBox("Admis AB")

	If note>=14 And note<16 Then MsgBox("Admis Bien")

	If note>=16 Then MsgBox("Admis TB")

End Sub



Exercice 7 (TD 3)



Soit 2 nombres x et y donnés par l'utilisateur, puis une lettre donnée par l'utilisateur. Si la lettre tapée est « p », le résultat calculé par le programme vaut le produit x*y ; si la lettre tapée est « s », le résultat calculé vaut la somme x+y.

Corrigé

Sub Operation()

	Dim op As String, x As Integer, y As Integer

	x=InputBox("x=")

	y=InputBox("y=")

	op=InputBox("operation ? p=produit s=somme")

	If op="p" Then

		MsgBox("Le produit de " & x & " et " & y & " vaut " & x*y)

	End If

	If op="s" Then

		MsgBox("La somme de " & x & " et " & y & " vaut " & x+y)

	End If

	If op<>"s" And op<>"p" Then

		' l'utilisateur a entré une lettre différente de "p" et "s"

		MsgBox("Erreur : opération inconnue !")

	End If

End Sub



Exercice 5 (TD 4)



Faire la somme de 10 nombres aléatoires : chaque nombre est le résultat d’un tirage au hasard, et la somme des nombres est calculée au fur et à mesure. Calculer ensuite la moyenne de ces nombres.

Corrigé

Sub Exo5TD4()

Dim somme As Double, x As Double, i As Integer

	Randomize

	somme=0

	For i=1 To 10 

		x=Rnd()

		somme=somme+x

	Next

	MsgBox("Moyenne = " & somme/10)

End Sub



Exercice 6 (TD 4)



Faire la somme de plusieurs nombres positifs en arrêtant ce calcul lors de la saisie de –1. Calculer ensuite la moyenne de ces nombres.



Corrigé

Sub Exo6TD4()

	Dim somme As Double, x As Double, n As Integer

	x=InputBox("Entrez un nombre réel")

	n=0 'la variable n sert à compter combien de 

	' nombres a entré l'utilisateur.

	somme=0

	While x<>-1

		somme=somme+x

		n=n+1

		x=InputBox("Entrez un nombre réel")

	Wend

	MsgBox("Moyenne des " & n & " nombres = " & somme/n)

End Sub





Exercice 3 (TD 5)



Modifier l'exercice précédent pour que maintenant le jeu s'arrête soit lorsque l'utilisateur répond "non" à une question posée soit lorsque le nombre de réalisations de ce jeu dépasse 10.



Corrigé

Sub Exo3TD5()

	Dim x As Integer, y As Integer, i As Integer, reponse As String

	Randomize

	i=0

	reponse="oui"

	While reponse<>"non"

		i=i+1

		x=Int(6*Rnd())+1 'tire au hasard un nombre entre 1 et 6

		y=InputBox("D'après vous, quel est le résultat du lancer au hasard ?")

		If x=y Then

			MsgBox("Vous avez gagné !")

		Else

			MsgBox("Vous avez perdu !")

		End If

		If i<10 Then

			reponse=InputBox("Voulez-vous continuer ?")

		Else

			MsgBox("Fini, vous n'avez droit qu'à 10 tentatives !")

			reponse="non"

		End If

	Wend

End Sub



Exercice  5 (TD 5)



On suppose un mot et une lettre saisis par l'utilisateur. Ecrire le programme qui vérifie si la lettre donnée est une lettre du mot. Le programme s'arrête dès qu'on a vérifié que la lettre est une lettre du mot.

N.B Pour connaitre la longueur du mot on utilisera len("mot") et pour obtenir la ième lettre mid("mot",i,1)

Corrigé

Sub Exo5TD5()

	Dim mot As String, lettre As String, i As Integer

	mot=InputBox("Entrez un mot")

	lettre=InputBox("Entrez une lettre")

	i=1

	While i<=len(mot) And mid(mot,i,1)<>lettre

		i=i+1

	Wend

'à la fin de la boucle, la variable i contient la position de

'la première occurrence de la lettre si le mot contient la lettre

'et prend la valeur len(mot)+1 si la lettre n'appartient pas au mot

	If i<=len(mot) Then 'la i-eme lettre du mot est la lettre cherchee

		MsgBox("Le mot " & mot & " contient la lettre " & lettre)

	Else

		MsgBox("Le mot " & mot & " ne contient pas la lettre " & lettre)	

	End If

End Sub

Exercice 1(TD 6)



"Pile ou face". L' utilisateur entre son choix qui est comparé au choix fait au hasard par l'ordinateur. Si les deux valeurs sont égales l'utilisateur a gagné. Programmer dix réalisations de ce jeu et afficher le score (un jeu gagné donne un point).



Corrigé

Sub PileFace()

	Dim x As Integer, p As String, q As String

	Dim score As Integer, i As Integer

	Randomize 'Initialisation du générateur de nombres aléatoires

	score=0

	For i=1 To 10

		x=Int(2*Rnd())+1 'tire au hasard un nombre entre 1 et 2

		If x=1 then

			p="pile"

		Else

			p="face"

		End If

		q=InputBox("Vous choisissez pile? Ou face?")

		If p=q Then 'L'utilisateur a deviné le résultat du lancer

			MsgBox("Vous avez gagné !")

			score=score+1 ' On augmente le score de un

		Else ' l'utilisateur n'a pas deviné le résultat du lancer

			MsgBox("Vous avez perdu  !")

		End If

	Next

	MsgBox("Vous avez gagné " & score & " fois.")

End Sub





Exercice 4 (TD 6)



Soit un mot donné par l'utilisateur, calculer le nombre de e de ce mot.



Corrigé

Sub Nombree()

	Dim mot As String, i As Integer, nb As Integer

	mot=InputBox("Entrez un mot")

	nb=0 'la variable nb sert à compter le nombre de e dans le mot

	For i=1 To len(mot)

		If mid(mot,i,1)="e" Then 'la i-ème lettre du mot est un "e"

			nb=nb+1

		End If

	Next

	MsgBox("nombre de e dans le mot " & mot & " = " & nb)

End Sub



Exercice 2 (TD 7)



Définir une fonction maximum(n1,n2,n3) qui retourne le maximum entre trois nombres. Utiliser ensuite cette fonction dans une macro  avec des nombres donnés par l’utilisateur. Utiliser cette fonction dans la feuille Excel.



Corrigé

' première version

Function max3(x As Integer, y As Integer, z As Integer) As Integer

	If x >= z And x >= y Then

		max3 = x

	ElseIf y >= z Then

		max3 = y

	Else

		max3 = z

	End If

End Function



' deuxième version

Function max3(x As Integer, y As Integer, z As Integer) As Integer

	If x >= z And x >= y Then max3 = x

	If y >= x And y >= z Then max3 = y

	If z >= y And z >= x Then max3 = z

End Function



' troisième version

Function max2(x As Integer, y As Integer) As Integer

	If x > y Then

		max2 = x

	Else

		max2 = y

	End If

End Function



Function max3(x As Integer, y As Integer, z As Integer) As Integer

	max3 = max2(x,max2(y,z))

End Function





Sub test1_max3()

	Dim x As Integer, y As Integer, z As Integer

	x = InputBox("Donner le premier nombre")

	y = InputBox("Donner le deuxième nombre")

	z = InputBox("Donner le troisième nombre")

	MsgBox ("Le maximum est : " & max3(x, y, z))

End Sub



Sub test2_max3()

	Dim x As Integer, y As Integer, z As Integer

	x = Range("A1").Value

	y = Range("A2").Value

	z = Range("A3").Value

	Range("A5") = max3(x, y, z)

End Sub

Sub test3_max3()

	Range("A5") = max3(Range("A1").Value,Range("A2").Value,Range("A3").Value)

End Sub

On peut aussi mettre =max3(A1;A2;A3) dans la cellule A5. La valeur de la cellule A5 sera alors automatiquement recalculée si on change une des trois cellules A1, A2 ou A3.



�Exercice 3 (TD 7)



Ecrire une fonction Perimetre(Largeur, Longueur) qui calcule le périmètre d’un rectangle. Ecrire une macro   qui demande à l’utilisateur d’entrer la largeur et la longueur d’un rectangle et affiche le périmètre.

Ecrire dans une feuille excel un tableau :

largeur�longueur��3cm�10cm��5cm�8cm��63cm�20cm��

Utiliser la fonction Perimetre pour calculer dans la colonne suivante le périmètre des rectangles sur chaque ligne.



Corrigé

Function perimetre(x As Integer, y As Integer) As Integer

	perimetre = 2 * x + 2 * y

End Function



�Sub test_perimetre()

	Dim x As Integer, y As Integer

	x = InputBox("donner la largeur du rectangle")

	y = InputBox("donnez la hauteur du ractangle")

	MsgBox ("le perimètre du rectangle est : " & perimetre(x, y))

End Sub



Function perimetre2(cel1 As Object, cel2 As Object) As Integer

'Cel1 et Cel2 sont deux cellules contenant des entiers

'la fonction peut être utilisée diretement dans la feuille excel

	Dim x As Integer, y As Integer

	x = cel1.Value

	y = cel2.Value

	perimetre2 = 2 * x + 2 * y

End Function



Exercice 1 (TD 8)



Soient 5 notes enregistrées au sein des cellules de la plage B1:B5. Définir la procédure qui calcule la moyenne des notes et la mention correspondante, et qui affecte les résultats aux cellules B6 et B7.

Définir une procédure qui colore dans une couleur (par exemple rouge = 3) les notes inférieures à 10.

Corrigé



Function mention(note As Double) As String ' première version

	If note < 10 Then mention = "refusé"

	If note >= 10 and note < 12 Then mention = "passable"

	If note >= 12 and note < 14 Then mention = "assez bien"

	If note >= 14 and note < 16 Then mention = "bien"

	If note >= 16 Then mention = "très bien"

End Function



Function mention(note As Double) As String ' deuxième version

	If note < 10 Then

		mention = "refusé"

	End If

	If note >= 10 and note < 12 Then

		mention = "passable"

	End If

	If note >= 12 and note < 14 Then

		mention = "assez bien"

	End If

	If note >= 14 and note < 16 Then

		mention = "bien"

	End If

	If note >= 16 Then

		mention = "très bien"

	End If

End Function



Function mention(note As Double) As String ' troisième version

	If note < 10 Then

		mention = "refusé"

	Else

		If note < 12 Then

			mention = "passable"

		Else

			If note < 14 Then

				mention = "assez bien"

			Else

				If note < 16 Then

					mention = "bien"

				Else

					mention = "très bien"

				End If

			End If

		End If

	End If

End Function



Function mention(note As Double) As String ' quatrième version

	If note < 10 Then

		mention = "refusé"

	ElseIf note < 12 Then

		mention = "passable"

	ElseIf note < 14 Then

		mention = "assez bien"

	ElseIf note < 16 Then

		mention = "bien"

	Else

		mention = "très bien"

	End If

End Function



Sub exo1()

	Dim maplage As Object

	Dim somme As Double, moyenne As Double, i As Integer

	Set maplage = Range("B1 : B5")

	somme = 0

	For i = 1 To maplage.count

		somme = somme + maplage.Cells(i).Value

	Next

	moyenne = somme / maplage.count

	Range("B6").Value = moyenne

	Range("B7").Value = mention(moyenne)

	'coloration en rouge des notes en dessous de 10

	For i = 1 To 5

		If maplage.Cells(i).Value < 10 Then

			maplage.Cells(i).Interior.ColorIndex = 3

		End If

	Next

End Sub



Exercice 3 (TD 8)



Soit une facture structurée par 5 lignes où figure sur chaque ligne, pour chacun des 5 produits, le nombre de produits et le prix unitaire. Définir la procédure qui calcule la somme à payer pour chaque produit ainsi que le montant total de la facture, et qui affecte les résultats aux cellules concernées.

Corrigé

Sub exo3()

	Dim factures As Object

	Dim somme As Double, i As Integer

	Set factures = Range("A22 : C26")

'la première colonne de la plage contient le prix unitaire du produit

'la deuxième colonne de la plage contient le quantité achetée du produit

'dans la troisème colonne on écrira le total à payer par produit

	somme = 0

	For i = 1 To 5

		factures.Cells(i, 3).Value = factures.Cells(i, 1).Value * factures.Cells(i, 2).Value

		somme = somme + factures.Cells(i, 3).Value

	Next

	'écriture de la somme à payer

	factures.Cells(6, 3).Value = somme

End Sub





Exercice 1 (TD 9)



Les prix de 5 articles sont mis dans les cellules B1:B5. Ecrire une procédure de deux paramètres taux1 et taux 2 qui permet de calculer  le montant total à payer sachant que ce prix est réduit selon un taux de réduction :

Si montant <500 alors pas de réduction

Si 500<=montant<1000 alors taux de réduction = taux 1

Si montant>=1000 alors taux de réduction = taux 2

Le programme doit alors affecter le résultat à la cellule B6.

Ecrire une procédure qui appelle cette macro avec les taux 0,02 et 0,05.



Modifier la macro pour qu’elle demande deux taux à l’utilisateur et les utilise pour calculer le montant total.

Corrigé

Sub addreduit(taux1 As Double, taux2 As Double)

	Dim r As Object, s As Double, i As Integer, v as double

	Set r = Range("B1:B5")

	For i = 1 To r.Count

		v = r.Cells(i).Value

		If v < 500 Then

			s = s + v

		ElseIf v < 1000 Then

			s = s + v * (1 - taux1)

		Else

			s = s + v * (1 - taux2)

		End If

	Next

	Range("b6") = s

End Sub

Sub reduc()

	Call addreduit(0.02, 0.05)

End Sub

Sub reduc2()

	Call addreduit(InputBox("taux1"), InputBox("taux2"))

End Sub



Exercice 2 (TD 9)



Un tableau de notes de 10 personnes de la forme suivante est placé dans une feuille Excel :

nom�Note1�Note2�Note3�Moyenne��dupond�12�07�3���durand�14�15�8���…�…�…�…���Sur la plage A1 :E11.



Complétez la colonne moyenne par une formule Excel adéquate.

Ecrire une macro qui permet, pour chaque personne de colorer en rouge toutes ses notes en dessous de 10 lorsque  sa moyenne est comprise entre 8 et 10.

Corrigé

Sub color810()

	Dim r As Object, i As Integer, j As Integer

	Set r = Range("B8:E17")

	For i = 1 To r.Rows.Count

		With r.Rows(i)

			If .Cells(.Count).Value >= 8 And .Cells(.Count).Value <= 10 Then

				For j = 1 To r.Columns.Count

					With .Cells(j)

						If .Value >= 8 And .Value <= 10 Then .Interior.Color = vbRed

					End With

				Next

			End If

		End With

	Next

End Sub



Exercice 3 (TD 9)



Ecrire une macro alter(coul1,coul2) qui colorie alternativement dans deux couleurs passées en paramètre une cellule sur deux de la première ligne de la plage Selection. On pourra utiliser la propriété Selection.Columns.Count qui donne le nombre de colonnes de la plage. Définir une macro test qui appelle la macro alter.



Ecrire une macro alterlg (coul1,coul2, lg) pour qu’elle colorie alternativement dans les deux la ligne de la plage Selection dont le numéro lg est passé en troisième paramètre. La tester.



Ecrire une macro damier(coul1,coul2) qui colorie en damier des deux couleurs passées en paramètre la plage de cellule Selection. Testez la macro.

Corrigé

Sub alter(couleur1 As Integer, couleur2 As Integer)

    Dim i As Integer

    For i = 1 To Selection.Columns.Count

        With Selection.Cells(i).Interior

            If i Mod 2 = 0 Then

                .ColorIndex = couleur1

            Else

                .ColorIndex = couleur2

            End If

        End With

    Next

End Sub

Sub alterrv()

    Call alter(3, 4)

End Sub

Sub damier(couleur1 As Integer, couleur2 As Integer)

    Dim i As Integer

    For i = 1 To Selection.Rows.Count

        For j = 1 To Selection.Columns.Count

            With Selection.Cells(i, j).Interior

                If (i + j) Mod 2 = 0 Then

                    .ColorIndex = couleur1

                Else

                    .ColorIndex = couleur2

                End If

            End With

        Next

    Next

End Sub

Sub damierrv()

    Call damier(3, 4)

End Sub

Exercice 1 (TD 10)



Ecrire une fonction prodcol qui calcule le produit scalaire de deux plages en colonne passées en paramètres (c'est-à-dire  si les valeurs de la première colonne sont a1,…ak, celles de la deuxième colonne b1,..,bk, la fonction calcule a1*b1+a2*b2+….+ak*bk.)

Rappel : si Maplage est un objet plage de cellules, Maplage.Rows.Count est le nombre de lignes de cette plage.



Un tableau de notes de 10 personnes de la forme suivante est placé dans une feuille Excel :

Coefficients�0.5�2�1.5���nom�Note1�Note2�Note3�Moyenne pondérée��dupond�12�07�3���durand�14�15�8���…�…�…�…���Comment utiliser la fonction prodcol pour calculer la moyenne pondérée de chaque personne dans la feuille de calcul?

Corrigé

Function prodcol(plage1 as object, plage2 as object) as double

	dim somme as double, i as integer

	somme = 0

	for i=1 to plage1.count

		somme=somme+plage1.cells(i).value*plage2.cells(i).value

	next

	prodcol = somme

end function



Supposons que la cellule devant contenir la moyenne pondérée de Dupont est la cellule E3. On    y met la formule :

=prodcol(A3:C3;A$1:C$1)/somme(A$1:C$1)

Puis on la recopie vers le bas.

Exercice 2 (TD 10)



Ecrire une fonction qui indique, à partir d’une plage de cellules contenant des nombres, la valeur la plus proche d’un nombre x dans cette plage.

Corrigé

Function plusproche(plage as object, x as double) as double

	dim i as integer, s as double, y as double

	s=plage.Cells(1).Value

	for i=2 to plage.Count

		y=plage.Cells(i)

		if abs(x-y)<abs(x-s) Then s=y

	next

	plusproche = s

end function



Sub test_plusproche()

    Dim x As Double, y As Double

    x = InputBox("nombre cherché")

    y = plusproche(Selection, x)

    MsgBox ("Le nombre le plus proche de " & x & " est " & y)

End Sub



Exercice 3 (TD 10)



Ecrire une fonction qui, à partir d’une plage de cellules et d’un nombre x, compte le nombre de cellules de valeur x dans la plage.



On dispose d’une plage de nombres entiers comprenant des valeurs entre 0 et 20. Ecrire une fonction qui calcule la valeur la plus fréquente.

Corrigé

Function nbocc(plage As Object, x As Integer) As Integer

    Dim i As Integer, n As Integer

    n = 0

    For i = 1 To plage.Count

        If plage.Cells(i).Value = x Then n = n + 1

    Next

    nbocc = n

End Function



Function plusfrequent(plage As Object) As Integer

    Dim fmax As Integer, i As Integer

    fmax = -1

    For i = 0 To 20

        n = nbocc(plage, i)

        If n > fmax Then

            fmax = n

            plusfrequent = i

        End If

    Next

End Function



Sub test_plusfrequent()

    Dim plage As Object, f As Integer

    Set plage = Selection

    f = plusfrequent(plage)

    MsgBox ("La note la plus fréquente est " & f)

End Sub

Exercice 4 (TD 10)



Reprendre la macro damier du TD n°9 et créez un bouton avec pour légende DAMIER dans votre feuille de calcul associé à un coloriage en damier de la plage sélectionnée.



Corrigé

Dans la feuille de calcul, dans le menu « affichage », sous-menu « barre d'outils », cochez la ligne « boîte à outils, contrôles » pour que la boîte à outils apparaisse sur l'écran.

Le premier bouton de cette boîte (avec une règle, une équerre et un crayon) permet de rentrer dans le mode « création » ou d'en sortir. Assurez vous d'être dans le mode création. (Le bouton doit vous proposer d'en sortir).

Cliquez sur le bouton représentant un bouton (sic), puis cliquez à l'endroit où vous voulez mettre ce nouveau bouton.

Cliquez sur le deuxième bouton de la boîte à outils (propriétés).

Changez le nom (name) : Mettez « dam ».

Changez la légende (caption)  : Mettez « damier ».

Cliquez sur le troisième bouton de la boîte à outils (visualisez le code). Complétez la procédure 

Private Sub dam_Click()

    Call damierrv

End Sub

(« damierrv » est la procédure que vous avez écrite au cours du TD 9).

Cliquez sur le premier bouton de la boîte à outils (avec une règle, une équerre et un crayon) pour sortir du mode « création ».

Sélectionnez une plage de données. Cliquez sur le bouton « damier ».

Si cela se passe bien, vous pouvez fermer toutes les fenêtres inutiles (pour éviter les courants d'air).




